



```
> restart
> with(LinearAlgebra):
  with(Physics):
  with(plots):
  with(plottools):
  with(DEtools):
  with(Typesetting):
  interface(typesetting=extended):
  interface(showassumed=0):
> x_m:=<x(t)+l*sin(theta(t)), -l*cos(theta(t))>:
print(`Mass m position vector`=x_m);

printf("\n");

v_m:=diff(x_m,t):
print(`Mass m velocity vector`= v_m);
```

$$\text{Mass } m \text{ position vector} = \begin{bmatrix} x(t) + l \sin(\theta(t)) \\ -l \cos(\theta(t)) \end{bmatrix}$$

$$\text{Mass } m \text{ velocity vector} = \begin{bmatrix} \dot{x}(t) + l \dot{\theta}(t) \cos(\theta(t)) \\ l \dot{\theta}(t) \sin(\theta(t)) \end{bmatrix}$$

```
> T_m:=1/2*m*(v_m(1)^2+v_m(2)^2):
V_m:=-m*g*l*cos(theta(t)):

printf("\n");

print(`Kinetic energy of mass m:`);
print('T__m=1/2*m*(v[x]^2+v[y]^2)');
T:=simplify(T_m):
print('T__m'=T_m);

printf("\n");

print(`Potential energy of mass m:`);
print('V__m'=V_m);

printf("\n");
```

*Kinetic energy of mass m:*

$$T_m = \frac{1}{2} m (\dot{v}_x^2 + \dot{v}_y^2)$$

$$T_m = \frac{m \left( (\dot{x}(t) + l \dot{\theta}(t) \cos(\theta(t)))^2 + l^2 \dot{\theta}(t)^2 \sin(\theta(t))^2 \right)}{2}$$

*Potential energy of mass m:*

$$V_m = -m g l \cos(\theta(t))$$

```
> T_M:=1/2*M*((diff(x(t),t))^2):
V_M:=1/2*k*x(t)^2:

printf("\n");

print(`Kinetic energy of mass M:`);
print('T_M'=T_M);

printf("\n");

print(`Potential energy of mass M:`);
print('V_M'=V_M);
```

*Kinetic energy of mass M:*

$$T_M = \frac{M \dot{x}(t)^2}{2}$$

*Potential energy of mass M:*

$$V_M = \frac{k x(t)^2}{2}$$

```
> T:=T_m+T_M:
V:=V_m+V_M:
L:=T-V:
print('L'=L);

eq1:=simplify(<diff(L,x(t)), diff(L,theta(t))>):
eq2:=simplify(<diff(L,diff(x(t),t)), diff(L,diff(theta(t),t))>):
eq3:=simplify(<diff(eq2[1],t), diff(eq2[2],t)>):

print('diff(L,x)'=eq1);
print('diff(L,diff(x,t))'=eq2);
print('diff(diff(L,diff(x,t)),t)'=simplify(eq3));

printf("\n");
print(`Thus,`);
print('diff(diff(L,diff(x,t)),t)=diff(L,x)');
eq4:=simplify({(eq3[1]-eq1[1]=0), (eq3[2]-eq1[2]=0)/(m*1)}):
print(<eq4[1],eq4[2]>);
eq5:=simplify(solve(eq4,{diff(x(t),t,t), diff(theta(t),t,t)})):
print(<eq5[1],eq5[2]>);
```

$$L = \frac{m \left( \dot{x}(t) + l \dot{\theta}(t) \cos(\theta(t)) \right)^2 + l^2 \dot{\theta}(t)^2 \sin^2(\theta(t))}{2} + \frac{M \dot{x}(t)^2}{2} + m g l \cos(\theta(t)) - \frac{k x(t)^2}{2}$$

$$\frac{\partial}{\partial x} L = \begin{bmatrix} -k x(t) \\ -m l \sin(\theta(t)) (\dot{\theta}(t) \dot{x}(t) + g) \end{bmatrix}$$

$$\frac{\partial}{\partial \dot{x}} L = \begin{bmatrix} (M+m) \dot{x}(t) + \dot{\theta}(t) \cos(\theta(t)) l m \\ m l (\cos(\theta(t)) \dot{x}(t) + \dot{\theta}(t) l) \end{bmatrix}$$

$$\frac{\partial^2}{\partial \dot{x} \partial t} L = \begin{bmatrix} (M+m) \ddot{x}(t) - m l (\dot{\theta}(t)^2 \sin(\theta(t)) - \ddot{\theta}(t) \cos(\theta(t))) \\ m l (-\dot{\theta}(t) \sin(\theta(t)) \dot{x}(t) + \cos(\theta(t)) \ddot{x}(t) + \ddot{\theta}(t) l) \end{bmatrix}$$

Thus,

$$\frac{\partial^2}{\partial \dot{x} \partial t} L = \frac{\partial}{\partial x} L$$

$$\begin{bmatrix} \ddot{\theta}(t) l + \cos(\theta(t)) \ddot{x}(t) + \sin(\theta(t)) g = 0 \\ (M+m) \ddot{x}(t) - \dot{\theta}(t)^2 \sin(\theta(t)) l m + \ddot{\theta}(t) \cos(\theta(t)) l m + k x(t) = 0 \end{bmatrix}$$

$$\begin{bmatrix} \ddot{\theta}(t) = \frac{-\dot{\theta}(t)^2 \cos(\theta(t)) \sin(\theta(t)) l m + x(t) \cos(\theta(t)) k - \sin(\theta(t)) g (M+m)}{l (-\cos(\theta(t))^2 m + M+m)} \\ \ddot{x}(t) = \frac{\dot{\theta}(t)^2 \sin(\theta(t)) l m + \sin(\theta(t)) g \cos(\theta(t)) m - k x(t)}{-\cos(\theta(t))^2 m + M+m} \end{bmatrix}$$

Let us take  $\theta(t) = \omega t$ , which means the small mass is rotating at a constant speed of  $\omega$ . Thus, we can safely take only the second equation and ignore the first equation.

```
> eq6:=theta(t)=omega*t:
eq7:=diff(eq6,t):
eq8:=subs({eq6,eq7},eq5[2]):
```

```
print(`Given:`):
print(eq6):
print(`Thus:`):
print(eq8):
print(`This is the final equation that we need to solve.`)
```

Given:

$$\theta(t) = \omega t$$

Thus:

$$\ddot{x}(t) = \frac{\omega^2 \sin(\omega t) l m + \sin(\omega t) g \cos(\omega t) m - k x(t)}{-\cos(\omega t)^2 m + M + m}$$

*This is the final equation that we need to solve.*

```
> OMEGA := 4:
GRAVITY := 9.8:
LENGTH := 0.02:
M:=10:
m:=1:
k:=10:
tf:=10:

> ode:=eval(eq8, {l=LENGTH,g=GRAVITY,omega=OMEGA}):
sol:=dsolve({ode, x(0)=0, D(x)(0)=0}, x(t), numeric, method=rkf45, output=
listprocedure):
x_array:=eval(x(t),sol):

big_ball := proc(x,y) plots[pointplot]([x,y],color=blue,symbol=solidcircle,
symbolsize=200): end proc:
small_ball := proc(x,y) plots[pointplot]([x,y],color=green,symbol=solidcircle,
symbolsize=30): end proc:

> anim1:=animate(big_ball, [x_array(t),0], t=0..tf,scaling=constrained,frames=100):
anim2:=animate(small_ball, [LENGTH*sin(OMEGA*t)+x_array(t), -LENGTH*cos(OMEGA*t)], t=0.
tf,scaling=constrained,frames=100):
anim3:=animate(plot,[[x_array(t),LENGTH*sin(OMEGA*t)+x_array(t)],[0, -LENGTH*cos(OMEGA*
t)],color=red], t=0..tf, scaling=constrained,frames=100):
anim4:=animate(plot,[[0, x_array(t)],[0, 0],color=yellow,thickness=5], t=0..tf,
scaling=constrained,frames=100):

h:=display(anim1,anim2,anim3,anim4);
```

$t = 10.0000000000$

